Newsgroup: it.comp.appl.access
Titolo: Come massimizzare la finestra report anteprima senza Docmd.Mazimise ?

Sto aprendo un report usando il comando
DoCmd.OpenReport  "rptDocumentoDoppio", acViewPreview, , "ID_DOC = 123456"
Reports!rptDocumentoDoppio.ZoomControl = 150
Come fare per aumentare al massimo la dimensione di una finestra anteprima 
report, senza usare il Docmd.Maximise ?
Vorrei evitare la massimizzazione perche' 'attacca' tutte le finestre una 
sopra l'altra
Vorrei invece mantenere la possibilita' di spostare le finestre all'intenro 
della cornice Access
Ciao e grazie
RobertoA


invece la puoi usare a patto che dopo fai un docmd.restore
-- 
-- 
MAssimiliano Amendola
www.accessgroup.it
Cisa - Conferenza Italiana Sviluppatori Access

 RobertoA
invece la puoi usare a patto che dopo fai un docmd.restore
-----------------------------------------
'la puoi fare dopo'  a che cosa ?
Il report deve stare la', non deve essere chiuso
E devo spostarmi sulle altre form, col report aperto
Ciao
RobertoA

> RobertoA
io ho un modulino che credo di aver prelevato dal sito di Alex:
------------------------------------
Option Compare Database
Option Explicit
Type Rect
    Left As Long
    Top As Long
    Right As Long
    Bottom As Long
End Type
Declare Function IsZoomed Lib "user32" (ByVal hWnd As Long) As Long
Declare Function GetWindowRect Lib "user32" (ByVal hWnd As Long, _
    lpRect As Rect) As Long
Declare Function ShowWindow Lib "user32" (ByVal hWnd As Long, _
    ByVal nCmdShow As Long) As Long
Declare Function MoveWindow Lib "user32" (ByVal hWnd As Long, _
    ByVal X As Long, ByVal Y As Long, ByVal nWidth As Long, _
    ByVal nHeight As Long, ByVal bRepaint As Long) As Long
Declare Function GetParent Lib "user32" (ByVal hWnd As Long) As Long
Const SW_MAXIMIZE = 3
Const SW_SHOWNORMAL = 1
Sub MaximizeRestoredForm(f As Form)
    Dim MDIRect As Rect
    Dim RetVal As Long
    ' If the form is maximized, restore it.
    If IsZoomed(f.hWnd) <> 0 Then
       RetVal = ShowWindow(f.hWnd, SW_SHOWNORMAL)
    End If
    ' Get the screen coordinates and window size of the
    ' MDIClient window.
    RetVal = GetWindowRect(GetParent(f.hWnd), MDIRect)
    ' Move the form to the upper-left corner of the MDIClient
    ' window (0,0) and size it to the same size as the
    ' MDIClient window.
    RetVal = MoveWindow(f.hWnd, 0, 0, MDIRect.Right - _
       MDIRect.Left - 4, MDIRect.Bottom - MDIRect.Top - 4, True)
End Sub
-------------------
e basta chiamare questa procedura con
MaximizeRestoredForm Me
oppure
MaximizeRestoredForm Forms("mascheraX")
e te la allarga al massimo della client area lasciandola restored!
IvanDaBologna
PS
per agire su un report devi cambiare il tipo di oggetto passato alla 
funzione da (f As Form) a (f As Report) e di conseguenza anche la chiamata:
MaximizeRestoredForm Reports("listafinale")
Probabilmente ti converrebbe 'doppiarla' con un altro nome per avere la 
caratteristica a disposizione per ambedue gli oggetti.

 ha scritto nel messaggio 
news:4934e5b7$0$41653$4faf...@reader4.news.tin.it...

Bingo !!!
Era proprio quello che cercavo
Grazie molte, ciao
RobertoA

 Sub MaximizeRestoredForm(f As Form)
L'originale è abbastanza vecchia e di Terry Kreft:
http://www.mvps.org/access/api/api0022.htm
Più tardi il suo codice è stata usata anche dalla gente del KB. p.e.
http://support.microsoft.com/kb/210123/en-us
-- 
cu
Karl
*********
Access FAQ: 
www.donkarl.com/it
 

 all'intenro della cornice Access
Però credo che se il bordo delle form è impostato a "Dialogo" il problema 
non sussista...

"Massimiliano Bertoglio" <max.be...@NOSPAM.it> ha scritto nel messaggio 
news:49351b1b$0$40311$4faf...@reader5.news.tin.it...

Verrebbe impedita solamente la massimizzazione della form/report
Se la form e' piccola rimarrebbe piccola
Io desidero invece ingrandire al massimo visibile l'anteprima report, 
indipendentemente dalla dimensione cornice di Access che puo' essere 
modificata dall'utente
Ciao
RobertoA